(2)

HDL-TM-89-16

February 1990

BASIC Graphics Routine Technical Manual

by Neal Tesny



U.S. Army Laboratory Command
Harry Diamond Laboratories
Adelphi, MD 20783-1197

DISCLAIMER NOTICE



THIS DOCUMENT IS BEST QUALITY AVAILABLE. THE COPY FURNISHED TO DTIC CONTAINED A SIGNIFICANT NUMBER OF PAGES WHICH DO NOT REPRODUCE LEGIBLY.

The findings in this report are not to be construed as an official Department of the Army position unless so designated by other authorized documents.

Citation of manufacturer's or trade names does not constitute an official endorsement or approval of the use thereof.

Destroy this report when it is no longer needed. Do not return it to the originator.

A preliminary version of this report was issued as HDL-PRL-89-9 in September 1989.

The findings in this report are not to be construed as an official Department of the Army position unless so designated by other authorized documents.

Citation of manufacturer's or trade names does not constitute an official endorsement or approval of the use thereof.

Destroy this report when it is no longer needed. Do not return it to the originator.

A preliminary version of this report was issued as HDL-PRL-89-9 in September 1989.

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

Form Approved OMB No. 0704-0188

1a. REPORT SECURITY CLASSIFICATION Unclassified	1b. RESTRICTIVE MARKINGS
2a. SECURITY CLASSIFICATION AUTHORITY	3. DISTRIBUTION/AVAILABILITY OF REPORT Approved for public release; distribution unlimited.
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE	
4. PERFORMING ORGANIZATION REPORT NUMBER(S) HDL-TM-89-16	5. MONITORING ORGANIZATION REPORT NUMBER(S)
6a. NAME OF PERFORMING ORGANIZATION Harry Diamond Laboratories 6b. OFFICE SYMBOL (*if applicable*) SLCHD-NW-ES	7a. NAME OF MONITORING ORGANIZATION
6c. ADDRESS (*City, State, and ZIP Code*) 2800 Powder Mill Road Adelphi, MD 20783-1197	7b. ADDRESS (*City, State, and ZIP Code*)
8a. NAME OF FUNDING/SPONSORING ORGANIZATION U.S. Army Laboratory Command 8b. OFFICE SYMBOL (*if applicable*) AMSLC	9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER
8c. ADDRESS (*City, State, and ZIP Code*) 2800 Powder Mill Road Adelphi, MD 20783-1145	10. SOURCE OF FUNDING NUMBERS PROGRAM ELEMENT NO. 6.20.21 / PROJECT NO. / TASK NO. / WORK UNIT ACCESSION NO.
11. TITLE (*Include Security Classification*) BASIC Graphics Routine Technical Manual	
12. PERSONAL AUTHOR(S) Neal Tesny	
13a. TYPE OF REPORT Final 13b. TIME COVERED FROM Mar 89 TO Aug 89	14. DATE OF REPORT (*Year, Month, Day*) February 1990 15. PAGE COUNT 34
16. SUPPLEMENTARY NOTATION AMS code: 6[illegible]0 HDL proj: XE79E5	
17. COSATI CODES FIELD / GROUP / SUB-GROUP	18. SUBJECT TERMS (*Continue on reverse if necessary and identify by block number*) graphics, BASIC, personal computer (PC)

19. ABSTRACT (*Continue on reverse if necessary and identify by block number*)

A new graphics routine in PC-BASIC for displaying waveforms is described. Linear plots are obtained in automatically scaled and labeled graticules. Specifiable options include window, viewport, and number of tic marks for each axis, and other options. The routine operates on any IBM-compatible PC. DOS is required with a version of the BASIC operating system. Comprehensive samples of output and a program listing are included.

20. DISTRIBUTION/AVAILABILITY OF ABSTRACT ☒ UNCLASSIFIED/UNLIMITED ☐ SAME AS RPT ☐ DTIC USERS	21. ABSTRACT SECURITY CLASSIFICATION Unclassified
22a. NAME OF RESPONSIBLE INDIVIDUAL Neal Tesny	22b. TELEPHONE (*Include Area Code*) (703) 490-2331 22c. OFFICE SYMBOL SLCHD-NW-ES

DD Form 1473, JUN 86 *Previous editions are obsolete.*

SECURITY CLASSIFICATION OF THIS PAGE

UNCLASSIFIED

Contents

Appendices

Tables

1. Introduction

The graphics capabilities of PC-BASIC are not sufficient to display data effectively. Low-level commands in BASIC such as DRAW, CIRCLE, and LINE[1] simply cannot be readily used to plot data. Because of the continual increase in the use of personal computers (PC's), and the need for readily available graphics in data processing, a sufficient plotting routine had to be written.

This routine results from years of experience with the handling and display of data on computers. It contains the options and versatility needed for most software plotting applications.

2. Description

2.1 General Description

This BASIC routine graphs an array/waveform on the terminal screen in an automatically generated and scaled graticule. It produces linear plots and uses engineering-unit scaling, i.e., scaling to powers of 10 in multiples of 3. It uses enhanced graphics adapter (EGA) BASIC screen-9 or -10 mode graphics, which give a 640 × 350 pixel resolution, or screen-2 or -8 mode graphics, which give a 640 × 200 pixel resolution.

In order to use this graphics routine, certain computer requirements must be filled. One must have a computer which is IBM-PC compatible, which has minimum processor requirements, such as that of an IBM-XT, and which has a version of the BASIC operating system. For nominal performance it is recommended that it have a coprocessor. In addition, the computer must have a graphics monitor and a graphics card such as Hercules, computer graphics adapter (CGA), or EGA, to name a few. In GW-BASIC, the CGA will only support screen-2 mode whereas the EGA will support screen-8, -9 and -10 modes. The monochrome graphics adapter (MGA) will only support screen 10 mode.

2.2 Options

Although all the routine needs from the user is an array, many options can be specified to control the appearance of the plotting. These options, listed in table 1, can be specified anywhere in the main program. The "aspects" of the graph which can be controlled are listed below.

1. Window—the minimum and maximum values, or "span," of the graph; includes the horizontal (x-axis) and vertical (y-axis) minimum and maximum values.

2. Viewport—the portion of the terminal screen the graph will be displayed in; i.e., the size and location of the graph on the screen are specified in terms of pixel coordinates.

3. Tics—number of divisions of the graticule; specified for the x-axis and the y-axis; the number of vertical and horizontal tics can be specified.

4. Labels—the numbers printed on the screen to signify the value of the plot at each tic mark, and the units of each axis which are printed; the number of digits the labels contain can be specified.

5. Colors—the colors of each of the following can be specified:

(a) Graticule—the grid the trace is displayed in.

(b) Trace—the "curve" drawn on the graticule, which represents the actual waveform.

(c) Labels—explained above.

An additional term which needs defining is pixel—the smallest unit of discernible graphics on the terminal screen; e.g., in screen-9 mode, the screen consists of 640 × 350 pixels.

If not specified, each option reverts to its default value. For example, if the window is not specified, it is autogenerated to give a trace which is full scale in the graticule and labels which are "nice, round" numbers. Not specifying the number of tics will have the routine automatically compute the correct number of tics to

---

[1] *Microsoft GW-BASIC Interpreter for the MS-DOS Operating System, Version 3.2, Microsoft Corporation (1986).*

Table 1. Options

Plot aspect	Range	Variable name	Descripton
Window	10^{-12} to 10^{15}	XWINDMIN	Minimum value the x-axis will have; default is 0
		XWINDMAX	Maximum value the x-axis will have; default is largest x-value of the trace
		YWINDMIN	Minimum value the y-axis will have
		YWINDMAX	Maximum value the y-axis will have
Viewpoint	1 to 639	XVIEWMIN	Leftmost horizontal pixel coordinate the graticule will have
		XVIEWMAX	Rightmost horizontal pixel coordinate the graticule will have
	1 to 349	YVIEWMIN	Vertical pixel coordinate the upper side of the graticule will have
		YVIEWMAX	Vertical pixel coordinate the lower side of the graticule will have
Tics	0 to 20	XTICS	Number of divisions to make in the graticule's x-axis
		YTICS	Number of divisions to make in the graticule's y-axis
Labels	1 to 8	XDIGITS	Number of digits the labels of the x-axis will have
		YDIGITS	Number of digits the labels of the y-axis will have
Colors	0 to 15	COLORGRA	Color of the graticule } –1 is black or no-draw on each
		COLORLAB	Color of the labels
		COLORTRA	Color of the trace

use, based upon the viewport and the window specifications. In addition, in case of errors or incorrectly entered values, default values are assigned. Also, options specified as noninteger are rounded to the nearest integer except for the window parameters and sampling interval.

There are also variables that are associated with the waveform, rather than with the plotting. These include the sampling interval, horizontal units, and vertical units. They do not have to be specified and are listed in table 2.

Sample outputs with the various options set are given in appendix A. These are hard copies, made directly from the terminal screen in screen-10 mode. In figures A-1 to A-10 the plots display the same waveform, which consists of a 512-element array with a sampling interval of 9.7656×10^{-8}. In figures A-11 through A-13 the plots display a 512-element array with a sampling interval of 1.9531×10^{-8}, along with the aforementioned waveform. Figure A-14 contains a plot of two different waveforms on the same graticule. Every waveform has the variables HORZUNIT$ AND VERTUNIT$ set to "S" and "V," respectively.

In order for the routine to function properly, a different scheme for printing numbers had to be devised. This is described in appendix B. Also, a pseudocode of the entire graphing routine is given in appendix C, and the program is listed in appendix D.

Table 2. Waveform Data

Component	Variable name	Description	Range
Sampling interval	SA	Default is 1	10^{-37} to 10^{38}
Horizontal units	HORZUNIT$	Default is no units displayed	10 characters
Vertical units	VERTUNIT$	Default is no units displayed	10 characters

3. Uses and Limitations

As stated earlier, in order to work the routine needs an array (in the proper array name, of course). The routine is called with the GOSUB command and it then plots the waveform on the screen. Although the routine only takes one array at a time, it can be used to plot many waveforms on the same graticule. This is easily done by first specifying the window and, second, by graphing each waveform separately, using the color options to omit redrawing of graticule and labels. An example of two waveforms displayed on the same graticule is shown in figure A-14 (app A).

In order to conserve memory in the computer, X-Y pairs are not used and only one array is assigned to a waveform. The method of X-Y pairs uses two arrays for each waveform. The first array holds the horizontal, or X-axis, value of each point and the second array contains the vertical, or Y-axis, value for each point. The two arrays are paired up to create X-Y pairs for each point of the waveform. In order to refrain from using X-Y pairs, one would use only the Y-value array but also have a sampling interval, a number which is the horizontal value between each point of the waveform. The X-axis value of a point is found by multiplying the sampling interval times the array-element number of the point of interest. The advantage of representing a waveform in this way is that one conserves memory in the computer (about one half the memory space is saved). The most significant disadvantage of representing a waveform in this way is, obviously, that the waveform must have a uniform sampling interval, and thus, nonuniformly sampled waveforms cannot be displayed.

However, if a waveform consists of several sections with different sampling intervals, it can be displayed with the routine. The procedure used is the same as that when plotting more than one waveform, except that each differently sampled portion is used in place of a separate waveform.

In general, this graphics routine has two applications: (1) it can be used simply to plot a waveform which is in a computer or on a disk or (2) it can be a building block for a data processing/acquisition package. It has enough modulicity so that it can be easily placed into a software package that is being written in BASIC. In fact it is anticipated that the routine will be used for this purpose in the near future.

Appendix A. Sample Output



Figure A-1. No options specified.



Figure A-2. YTICS=10, YWINDMIN=-.6, YWINDMAX=.6.

APPENDIX A



Figure A-3. YWINDMIN=-1.13, YWINDMAX=1.17.



Figure A-4. YWINDMIN=-.543, YWINDMAX=.577



Figure A-5. XWINDMIN=0, XWINDMAX=1E-05.



Figure A-6. XWINDMIN=4.829E-06, XWINDMAX=1.725E-05.

APPENDIX A



Figure A-7. COLORLAB=–1.



Figure A-8. COLORLAB=–1, COLORGRA=–1.



Figure A-9. YDIGITS=5.



Figure A-10. YDIGITS=5, XDIGITS=4.

APPENDIX A



Figure A-11. XVIEWMIN=50, XVIEWMAX=300; First Waveform: YVIEWMIN=35, YVIEWMAX=140; Second Waveform: YVIEWMIN=204, YVIEWMAX=309.



Figure A-12. YWINDMIN=-.543, WWINDMAX=.577, XVIEWMIN=50, XVIEWMAX=300; First Waveform: XVIEWMIN=35, XVIEWMAX=140; Second Waveform: YVIEWMIN=204, YVIEWMAX=309.





Figure A-13. YWINDMIN=-.6, YWINDMAX=.6, XVIEWMIN=50, XVIEWMAX=300; First Waveform: XVIEWMIN=35, XVIEWMAX=140; Second Waveform: YVIEWMIN=204, YVIEWMAX=309.



Figure A-14. Two waveforms.

Appendix B.—The Use of Graphically Produced Numbers

In GW-BASIC, one can only print characters in certain discrete places on the screen, namely in an 80 × 25 grid. This is a major weakness of IBM-type BASIC—characters cannot be printed where desired. If one tries printing numbers to line up with graticule tic marks, their positions come out distorted. See figure B-1.

A remedy to this is to draw the numbers individually. To do this the numbers must first be designed, each in a 14 × 8 pixel grid, the size of the regularly printed numbers. For ample space between the numbers when they are printed consecutively, the numbers must be made small enough inside the 14 × 8 grid. Examples of number designs are shown in figure B-2. The numbers used here are 9 pixels high by 6 pixels wide.

One of these numbers is printed when we move the graphics cursor to any of the 350- × 640-element locations at which to begin printing the number and use the DRAW command to highlight the appropriate pixel pattern on the screen. With the appropriate graphic string for each digit, an entire line of numbers can be printed.



Figure B-1. "Distorted" labels.

APPENDIX B



Figure B-2. Examples of graphically produced numbers.

Appendix C.—Pseudocode

Round option-variables to nearest integer except sampling interval and window parameters.

See if each value is within right limits.

Assign default values to the appropriate option parameters.

Auto-compute window and number of tic marks, if needed.

Compute default number of digits in the labels, if needed.

Compute labels.

Draw graph:

    Draw graph border

    Draw tic marks

    Print labels (use graphically produced numbers described in app B)

    Plot trace:

        For I = 1 to WAVEFORM_SIZE

            Do a matrix conversion from each point's amplitude to screen coordinates

            Draw a line to connect each point of the waveform

        Next I

End

Appendix D. Program Listing

```
10 REM ********* GRAPHME.BAS *********
20 REM 4/89 N TESNY
30 REM THIS IS THE MAIN PROGRAM WHICH CALLS THE GRAPHING ROUTINE.
40 WFSIZE=1000
50 OPEN "I",1,"FILE.DAT"
60 DIM WAVEFORM(WFSIZE)
70 SA=1E-09
80 HORZUNIT$="S"
90 VERTUNIT$="V"
100 YMINIMUM=1E+20:YMAXIMUM=-1E+20
110 FOR I= 1 TO WFSIZE
120   INPUT #1,WAVEFORM(I)
130   IF WAVEFORM(I)<YMINIMUM THEN YMINIMUM=WAVEFORM(I)
140   IF WAVEFORM(I)>YMAXIMUM THEN YMAXIMUM=WAVEFORM(I)
150 NEXT I
160 CLOSE #1
170 SCREEN 9
180 KEY OFF
190 CLS
200 GOSUB 5000
210 END
5000 REM ********* GRAPHME1.BAS *********
5010 REM 4/89 N TESNY
5020 REM GRAPHICS ROUTINE
5030 REM TAKE CINT OF ALL VARS, FIND DEFAULT VALUES:.....
5040 XVIEWMIN=CINT(XVIEWMIN):XVIEWMAX=CINT(XVIEWMAX)
5050 YVIEWMIN=CINT(YVIEWMIN):YVIEWMAX=CINT(YVIEWMAX)
5060 XWINDMN=0:XWINDMX=0:YWINDMN=0:YWINDMX=0
5070 IF XWINDMAX>XWINDMIN THEN XWINDMN=XWINDMIN:XWINDMX=XWINDMAX
5080 IF YWINDMAX>YWINDMIN THEN YWINDMN=YWINDMIN:YWINDMX=YWINDMAX
5090 XNTICS=CINT(XTICS):IF XNTICS>20 OR XNTICS<0 THEN XNTICS=0
5100 YNTICS=CINT(YTICS):IF YNTICS>20 OR YNTICS<0 THEN YNTICS=0
5110 XNDIGITS=CINT(XDIGITS):IF XNDIGITS<0 OR XNDIGITS>8 THEN XNDIGITS=0
5120 YNDIGITS=CINT(YDIGITS):IF YNDIGITS<0 OR YNDIGITS>8 THEN YNDIGITS=0
5130 SAMPINTV=SA:IF SAMPINTV=0 THEN SAMPINTV=1
5140 COLOGRAT=CINT(COLORGRA):IF COLOGRAT=0 OR COLOGRAT>15 THEN COLOGRAT=7
5150 COLOTRAC=CINT(COLORTRA):IF COLOTRAC=0 OR COLOTRAC>15 THEN COLOTRAC=12
5160 COLOLABE=CINT(COLORLAB):IF COLOLABE=0 OR COLOLABE>15 THEN COLOLABE=15
5170 IF XVIEWMIN>=XVIEWMAX OR XVIEWMIN<0 OR XVIEWMAX>640 THEN XVIEWMIN=80:XVIEWM
AX=560
5180 IF YVIEWMIN>=YVIEWMAX OR YVIEWMIN<0 OR YVIEWMAX>350 THEN YVIEWMIN=36:YVIEWM
AX=306
5190 REM COMPUTE WINDOW FOR GRAPH
5200 REM X-AXIS:
5210 XMAXIMUM=SAMPINTV*WFSIZE
5220 IF XNTICS=0 THEN XNTICS=CINT((XVIEWMAX-XVIEWMIN)/(6*8))
5230 IF XWINDMN>=XWINDMX THEN XWINDMN=0:XWINDMX=XMAXIMUM
5240 XSPAN=(XWINDMX-XWINDMN)/XNTICS
5250 REM COMPUTE XNDIGITS:
5260 IF XNDIGITS<>0 OR XSPAN=0 THEN 5290
5270 XNDIGITS=3:IF XWINDMN/XSPAN>100 AND XWINDMX/XSPAN>100 THEN XNDIGITS=4
```

APPENDIX D

```
5280 REM Y-AXIS:
5290 NTICNOM=CINT((YVIEWMAX-YVIEWMIN)/(2*14))
5300 MAX=YMAXIMUM:MIN=YMINIMUM
5310 IF MIN<>MAX OR YWINDMN<YWINDMX THEN 5360
5320 IF YNTICS<>0 THEN NTICNOM=YNTICS
5330 MIN=MIN*(1-NTICNOM/200)
5340 MAX=MAX*(1+NTICNOM/200)
5350 IF MIN=0 THEN MIN=-.000001*INT(NTICNOM/2):MAX=-.000001*INT(-NTICNOM/2)
5360 IF YWINDMN>=YWINDMX AND YNTICS<>0 THEN 5430
5370 IF YWINDMN< YWINDMX AND YNTICS= 0 THEN 5550
5380 IF YWINDMN>=YWINDMX AND YNTICS= 0 THEN 5640
5390 REM MODULE 1: REM TICS AND WINDOW DEFINED
5400 YSPAN=(YWINDMX-YWINDMN)/YNTICS
5410 GOTO 5790
5420 REM MODULE 2: WINDOW NOT DEFINED, #TICS SPECIFIED
5430 Y=(MAX-MIN)/YNTICS
5440 K=0
5450 REM ROUND Y UP TO NEAREST INTEGER:
5460 IF Y>10 THEN Y=Y/10:K=K+1:GOTO 5460
5470 IF Y<1 THEN Y=Y*10: K=K-1:GOTO 5470
5480 Y=-INT(-Y)
5490 YSPAN=10^K * Y
5500 YWINDMN=YSPAN*INT(MIN/YSPAN)
5510 YWINDMX=YWINDMN+YNTICS*YSPAN
5520 IF YWINDMX<MAX THEN Y=Y+1:GOTO 5460
5530 GOTO 5790
5540 REM MODULE 3: WINDOW DEFINED, TICS NOT SPECIFIED
5550 IF NTICNOM<=5 THEN 5600
5560 FOR YNTICS=NTICNOM TO (NTICNOM-2) STEP -1
5570   YSPAN=(YWINDMX-YWINDMN)/YNTICS
5580   IF YSPAN=INT(YSPAN) THEN 5610
5590 NEXT YNTICS
5600 YNTICS=NTICNOM
5610 YSPAN=(YWINDMX-YWINDMN)/YNTICS
5620 GOTO 5790
5630 REM MODULE 4: WINDOW AND #TICS NOT SPECIFIED
5640 Y (MAX-MIN)/NTICNOM
5650  =0
5660 REM ROUND Y UP TO NEAREST 1,2, OR 5:
5670 IF Y>10 THEN Y=Y/10:K=K+1:GOTO 5670
5680 IF Y<1 THEN Y=Y*10: K=K-1:GOTO 5680
5690 Y=-INT(-Y)
5700 IF Y=1 OR Y=2 OR Y=5 OR Y=10 THEN 5730
5710 Y=Y+1 : GOTO 5690
5720 REM FIND YNTICS, YSPAN:
5730 YSPAN=10^K * Y
5740 YWINDMN=YSPAN*INT(MIN/YSPAN)
5750 YWINDMX=YSPAN*-INT(-MAX/YSPAN)
5760 YNTICS=(YWINDMX-YWINDMN)/YSPAN
5770 IF YNTICS > NTICNOM THEN Y=Y+1:GOTO 5670
5780 REM COMPUTE YNDIGITS:
```

```
5790 IF YNDIGITS<>0 OR YSPAN=0 THEN 5820
5800 YNDIGITS=3:IF YWINDMN/YSPAN>100 AND YWINDMX/YSPAN>100 THEN YNDIGITS=4
5810 YNTICS=CINT(YNTICS)
5820 REM THE ACTUAL WAVEFORM PLOTTING
5830 IF COLOGRAT<0 THEN 6070
5840 REM DRAW GRAPH BORDER:
5850 DRAW "C"+STR$(COLOGRAT)
5860 X$=STR$(XVIEWMIN)+","+STR$(YVIEWMAX)
5870 DRAW "BM "+X$
5880 DRAW "M +"+STR$(XVIEWMAX-XVIEWMIN)+",0"
5890 DRAW "M +0,"+STR$(YVIEWMIN-YVIEWMAX)
5900 DRAW "M -"+STR$(XVIEWMAX-XVIEWMIN)+",0"
5910 DRAW "M +0,"+STR$(YVIEWMAX-YVIEWMIN)
5920 REM DRAW X-UNIT TIC MARKS:
5930 FOR I=1 TO XNTICS-1
5940   X=XVIEWMIN+(XVIEWMAX-XVIEWMIN)*I/XNTICS
5950   X=CINT(X)
5960   DRAW "BM "+STR$(X)+","+STR$(YVIEWMIN)
5970   DRAW "NM +0,"+STR$(YVIEWMAX-YVIEWMIN)
5980 NEXT I
5990 REM DRAW Y-UNIT TIC MARKS:
6000 FOR I=1 TO YNTICS-1
6010   Y=YVIEWMIN+(YVIEWMAX-YVIEWMIN)*I/YNTICS
6020   Y=CINT(Y)
6030   DRAW "BM "+STR$(XVIEWMIN)+","+STR$(Y)
6040   DRAW "NM +"+STR$(XVIEWMAX-XVIEWMIN)+",0"
6050 NEXT I
6060 REM COMPUTE SCALING FACTORS FOR ENGIN. UNITS FOR GRAPH
6070 X=XWINDMX
6080 GOSUB 6760
6090 XEXPON$=EXPONENT$ : XFACTOR=10^X
6100 X=YWINDMX
6110 Y=ABS(YWINDMN):IF Y>X THEN X=Y
6120 GOSUB 6760
6130 YEXPON$=EXPONENT$ : YFACTOR=10^X
6140 REM PRINT X-LABELS:
6150 IF COLOLABE<0 THEN 6540
6160 DRAW "C"+STR$(COLOLABE)
6170 AXISTYPE=1:NDIGITS=XNDIGITS
6180 NDIGMAX=3
6190 I=XWINDMX
6200 IF ABS(XWINDMN)>I THEN I=ABS(XWINDMN)
6210 IF I/XFACTOR<100 THEN NDIGMAX=2
6220 IF I/XFACTOR<10 THEN NDIGMAX=1
6230 NUMBER=XWINDMN/XFACTOR
6240 X=XVIEWMIN:Y=YVIEWMAX
6250 GOSUB 6880:REM PRINT LABEL
6260 FOR T=1 TO XNTICS
6270   X=XVIEWMIN+(XVIEWMAX-XVIEWMIN)*T/XNTICS
6280   X=CINT(X)
6290   NUMBER=(XWINDMN+XSPAN*T)/XFACTOR
```

APPENDIX D

```
6300   GOSUB 6880:REM PRINT LABEL
6310 NEXT T
6320 Y=Y/14+2:IF Y>25 THEN Y=25
6330 LOCATE Y,(XVIEWMAX+XVIEWMIN)/16
6340 PRINT XEXPON$;HORZUNIT$;:LOCATE 1,25
6350 REM PRINT Y-LABELS:
6360 AXISTYPE=2:NDIGITS=YNDIGITS
6370 NDIGMAX=3
6380 I=YWINDMX
6390 IF ABS(YWINDMN)>I THEN I=ABS(YWINDMN)
6400 IF I/YFACTOR<100 THEN NDIGMAX=2
6410 IF I/YFACTOR<10 THEN NDIGMAX=1
6420 NUMBER=YWINDMN/YFACTOR
6430 X=XVIEWMIN:Y=YVIEWMAX
6440 GOSUB 6880:REM PRINT LABEL
6450 FOR T=1 TO YNTICS
6460   Y=YVIEWMAX-(YVIEWMAX-YVIEWMIN)*T/YNTICS
6470   Y=CINT(Y)
6480   NUMBER=(YWINDMN+YSPAN*T)/YFACTOR
6490   GOSUB 6880:REM PRINT LABEL
6500 NEXT T
6510 X=XVIEWMIN/8-LEN(YEXPON$+VERTUNIT$):IF X<1 THEN X=1
6520 Y=YVIEWMIN/14-1:IF Y<1 THEN Y=1
6530 LOCATE Y,X:PRINT YEXPON$;VERTUNIT$
6540 REM PLOT TRACE:
6550 IF COLOTRAC<0 THEN 6730
6560 DRAW "C"+STR$(COLOTRAC)
6570 ARRAYPT1=CINT(XWINDMN/SAMPINTV)
6580 ARRAYPT2= INT(XWINDMX/SAMPINTV)
6590 IF ARRAYPT1<1 THEN ARRAYPT1=1 : STARTPOS=1
6600 IF ARRAYPT2>WFSIZE THEN ARRAYPT2=WFSIZE
6610 J=XWINDMX-XWINDMN:K=XVIEWMAX-XVIEWMIN
6620 M=YWINDMX-YWINDMN:N=YVIEWMAX-YVIEWMIN
6630 X=XVIEWMIN+(ARRAYPT1*SAMPINTV-XWINDMN)/J*K
6640 Y=YVIEWMAX-(WAVEFORM(ARRAYPT1)-YWINDMN)/M*N
6650 X=CINT(X):Y=CINT(Y)
6660 DRAW "BM "+STR$(X)+","+STR$(Y)
6670 FOR I=ARRAYPT1+1 TO ARRAYPT2
6680   X=XVIEWMIN+(I*SAMPINTV-XWINDMN)/J*K
6690   Y=YVIEWMAX-(WAVEFORM(I)-YWINDMN)/M*N
6700   X=CINT(X):Y=CINT(Y)
6710   DRAW "M "+STR$(X)+","+STR$(Y)
6720 NEXT I
6730 RETURN
6740 END
6750 REM SUBROUTINE TO COMPUTE SCALING FACTORS FOR ENGINEERING UNITS
6760 IF X>=1E+12 THEN EXPONENT$="1E+12 ":X=12:RETURN
6770 IF X>=1E+09  THEN EXPONENT$="1E+9 ":X=9 : RETURN
6780 IF X>=1000000!  THEN EXPONENT$="1E+6 ": X=6 : RETURN
6790 IF X>=1000!  THEN EXPONENT$="1E+3 ": X=3: RETURN
6800 IF X>=1!  THEN EXPONENT$=" ": X=0: RETURN
```

```
6810 IF X<1E-09  THEN EXPONENT$="1E-12 ": X=-12: RETURN
6820 IF X<.000001  THEN EXPONENT$="1E-9 ":X=-9: RETURN
6830 IF X<.001  THEN EXPONENT$="1E-6 ": X=-6: RETURN
6840 IF X<1      THEN EXPONENT$="1E-3 ": X=-3 : RETURN
6850 RETURN
6860 REM LABEL ROUTINE:************************************************************
6870 REM ROUNDING ROUTINE
6880 IF NDIGITS<=4 THEN NUMBER=CINT(NUMBER*10^(NDIGITS-NDIGMAX))/10^(NDIGITS-NDI
GMAX)
6890 REM PADDING ROUTINE:
6900 NUMBER$=STR$(NUMBER)
6910 IF INSTR(NUMBER$," ")=1 THEN NUMBER$=RIGHT$(NUMBER$,LEN(NUMBER$)-1)
6920 IF NDIGMAX>=NDIGITS THEN 7020
6930 L=LEN(NUMBER$)
6940 J=INSTR(NUMBER$,".")
6950 IF J=1 THEN NUMBER$="0"+NUMBER$
6960 IF J=0 THEN NUMBER$=NUMBER$+".": J=L
6970 FOR I=1 TO (NDIGITS-NDIGMAX)-(L-J)
6980   NUMBER$=NUMBER$+"0"
6990 NEXT I
7000 IF L-J>NDIGITS-NDIGMAX THEN NUMBER$=LEFT$(NUMBER$,L-(L-!)+(NDIGITS-NDIGMAX)
)
7010 REM PRINTING ROUTINE:
7020 LENPIXS=8*LEN(NUMBER$)
7030 IF INSTR(NUMBER$,".")>0 THEN LENPIXS=LENPIXS-3: REM A "." ONLY 5 PIXELS LON
G
7040 REM FIND STARTING POINT:
7050 IF AXISTYPE=1 THEN XSTART=X-INT((LENPIXS-1)/2): YSTART=Y+12
7060 IF AXISTYPE=2 THEN XSTART=X-LENPIXS-3: YSTART=Y+4
7070 REM IF LABEL IS OFFSCREEN THEN SLIDE IT ONSCREEN:
7080 IF XSTART<0 THEN XSTART=0
7090 IF YSTART>350 THEN YSTART=350
7100 REM PRINT NUMBER$:
7110 DRAW "BM"+STR$(XSTART)+","+STR$(YSTART)
7120 FOR I=1 TO LEN(NUMBER$)
7130   DIGIT$=MID$(NUMBER$,I,1)
7140 IF DIGIT$="0" THEN DRAW "BR5 L3 H U6 E R3 F D6 BF BR":GOTO 7260
7150 IF DIGIT$="." THEN DRAW "BR U R D BR2" :GOTO 7260
7160 IF DIGIT$="-" THEN DRAW "BU4 BR R5 BD4 BR2":GOTO 7260
7170 IF DIGIT$="1" THEN DRAW "BU6 BR3 E U D8 L2 R4 BR2" :GOTO 7260
7180 IF DIGIT$="2" THEN DRAW "BU7 BR E R3 F D G5 D R5 BR2"
7190 IF DIGIT$="3" THEN DRAW "BE F R3 E U2 H L3 R3 E U2 H L3 G BR7 BD7"
7200 IF DIGIT$="4" THEN DRAW "BR5 U8 G4 D1 R5 BD3 BR2"
7210 IF DIGIT$="5" THEN DRAW "BE F R3 E U2 H L4 U4 R5 BR2 BD8"
7220 IF DIGIT$="6" THEN DRAW "BU4 BR R4 F D2 G L3 H U6 E R3 F BR2 BD7"
7230 IF DIGIT$="7" THEN DRAW "BR BU8 R5 D2 G3 D3 BR5"
7240 IF DIGIT$="8" THEN DRAW "BU3 BR D2 F R3 E U2 H L3 H U2 E R3 F D2 BR2 BD5"
7250 IF DIGIT$="9" THEN DRAW "BE F R3 E U2 H L3 H U2 E R3 F D3 BR2 BD4"
7260 NEXT I
7270 RETURN
```

DISTRIBUTION

ADMINISTRATOR
DEFENSE TECHNICAL INFORMATION CENTER
ATTN DTIC-DDA (12 COPIES)
CAMERON STATION, BUILDING 5
ALEXANDRIA, VA 22304-6145

DIRECTOR
US ARMY BALLISTIC RESEARCH LABORATORY
ATTN SLCBR-DD-T (STINFO)

US ARMY ELECTRONICS TECHNOLOGY
& DEVICES LABORATORY
ATTN SLCET-DD
FT MONMOUTH, NJ 07703

DIRECTOR
US ARMY MATERIEL SYSTEMS ANALYSIS
ACTIVITY
ATTN AMXSY-MP
ABERDEEN PROVING GROUND, MD 21005

COMMANDER
US ARMY MISSILE & MUNITIONS
CENTER & SCHOOL
ATTN ATSK-CTD-F
REDSTONE ARSENAL, AL 35809

US ARMY BELVOIR RD&E CENTER
(FORMERLY US ARMY MOBILITY EQUIPMENT
RESEARCH & DEVELOPMENT COMMAND)
ATTN STRBE-F
ATTN STRBE-FG
FT BELVOIR, VA 22060

DOD
PM MOBILE ELECTRIC POWER
(FORMERLY US ARMY MOBILITY EQUIPMENT
RESEARCH & DEVELOPMENT COMMAND)
ATTN AMC-PM-MEP-T
7500 BACKLICK ROAD
BUILDING 2089
SPRINGFIELD, VA 22150-3107

COMMANDER
(FORMERLY US ARMY MOBILITY EQUIPMENT
RESEARCH & DEVELOPMENT COMMAND)
US ARMY ENGINEER SCHOOL
ATTN ATZA-TSM-G
FT BELVOIR, VA 22060-5249

HQ, USAF/SAMI
WASHINGTON, DC 20330

ANALYTICAL SYSTEMS ENGINEERING CORP
OLD CONCORD ROAD
ATTN LIBRARIAN
BURLINGTON, MA 01803

BROOKHAVEN
DEPT OF ENERGY
ASSOCIATED UNIVERSITIES, INC
ATTN TECHNICAL INFORMATION DIV
ATTN PHYSICS DEPT, 5103
UPTON, LONG ISLAND, NY 11973

DEPARTMENT OF COMMERCE
NATIONAL BUREAU OF STANDARDS
ATTN LIBRARY
WASHINGTON, DC 20234

NATIONAL OCEANIC & ATMOSPHERIC ADM
ENVIRONMENTAL RESEARCH LABORATORIES
ATTN LIBRARY, R-51, TECH REPORTS
BOULDER, CO 80302

DIRECTOR
DEFENSE COMMUNICATIONS ENGINEERING
CENTER
ATTN TECHNICAL LIBRARY
1860 WIEHLE AVE
RESTON, VA 22090

DIRECTOR
NATIONAL SECURITY AGENCY
ATTN TECHNICAL LIBRARY
FT MEADE, MD 20755

DIRECTOR
DEFENSE NUCLEAR AGENCY
ATTN TISI, SCIENTIFIC INFORMATION DIV
ATTN TITL, TECHNICAL LIBRARY DIV
WASHINGTON, DC 20305

COMMANDER
US ARMY ABERDEEN PROVING GROUND
ATTN STEAP-TL, TECH LIB
ABERDEEN PROVING GROUND, MD 21005

DIRECTOR
APPLIED TECHNOLOGY LABORATORY
AVRADCOM
ATTN DAVDL-ATL-TSD, TECH LIBRARY
FT EUSTIS, VA 23604

COMMANDER/DIRECTOR
CHEMICAL SYSTEMS LABORATORY
US ARMY ARMAMENT MUNITIONS &
CHEMICAL COMMAND
US ARMY ARMAMENT RESEARCH & DEVELOPMENT
CENTER
ATTN DRDAR-CLJ-L, TECHNICAL LIBRARY BR
ABERDEEN PROVING GROUND, MD 21010

DISTRIBUTION (cont'd)

COMMANDER
US ARMY COMM-ELEC ENGR INSTAL AGENCY
ATTN TECH LIB
FT HUACHUCA, AZ 85613

COMMANDER
US ARMY COMPUTER SYSTEMS COMMAND
ATTN TECH LIB
FT BELVOIR, VA 22060

COMMANDER
TECHNICAL SUPPORT ACTIVITY
ATTN DELSD-L, TECH LIB DIR
FT MONMOUTH, NJ 07703

US ARMY LABORATORY COMMAND
ATTN TECHNICAL DIRECTOR, AMSLC-TD

INSTALLATION SUPPORT ACTIVITY
ATTN LEGAL OFFICE, SLCIS-CC

USAISC
ATTN RECORD COPY, AMSLC-IM-TS
ATTN TECHNICAL REPORTS BRANCH,
AMSLC-IM-TR (2 COPIES)

HARRY DIAMOND LABORATORIES
ATTN D/DIVISION DIRECTORS
ATTN LIBRARY, SLCHD-TL (3 COPIES)
ATTN LIBRARY, SLCHD-TL (WOODBRIDGE)
ATTN CHIEF, SLCHD-NW-CS
ATTN CHIEF, SLCHD-NW-E
ATTN CHIEF, SLCHD-NW-EH
ATTN CHIEF, SLCHD-NW-EP
ATTN CHIEF, SLCHD-NW-ES
ATTN CHIEF, SLCHD-NW-P
ATTN CHIEF, SLCHD-NW-R
ATTN CHIEF, SLCHD-NW-RP
ATTN CHIEF, SLCHD-NW-RS
ATTN CHIEF, SLCHD-NW-TN
ATTN CHIEF, SLCHD-NW-TS
ATTN N. TESNY, SLCHD-NW-ES (20 COPIES)